SENSITIVITY ANALYSIS FOR MULTI-ATTRIBUTE UTILITY USING EXCEL

Michael R. Middleton, School of Business and Management, University of San Francisco,
2130 Fulton Street, San Francisco, CA, 94117-1045, 415 422 6768, middleton@usfca.edu

This paper describes several standard methods for analyzing decisions where the outcomes have multiple attributes. The example problem concerns a large company that is planning to purchase several hundred cars for use by the sales force. The company wants a car that is inexpensive, safe, and lasts a long time. Figure 1 shows data for seven cars that are being considered.

Figure 1 Attribute Data for Seven Alternatives

	A	B	C	D	E	F	G	H
1		Alternatives						
2	Attribute	Alta	Bulldog	Cruiser	Delta	Egret	Fleet	Garnett
3	Cost	$20	$18	$16	$14	$12	$10	$15
4	Lifetime	10	10	8	8	6	6	8
5	Safety	High	Medium	High	Medium	Medium	Low	Low
6								
7	Cost	thousands of dollars						
8	Lifetime	expected years						
9	Safety	third-party rating						

Other attributes might be important, e.g., comfort and prestige. The cost attribute should include operating costs, insurance, and salvage value, in addition to purchase price. It might be appropriate to combine the cost and lifetime attributes into a single attribute, e.g., cost per year. Clemen [1] suggests that a set of attributes should be complete (so that all important objectives are included), as small as possible (to facilitate analysis), not redundant (to avoid double-counting a common underlying characteristic), and decomposable (so that the decision maker can think about each attribute separately).

Dominance

An alternative can be eliminated if another alternative is better on some objectives and no worse on the others. The Garnett is more expensive than the Delta, has the same lifetime, and has a lower safety rating. So the Garnett can be eliminated from further consideration.

Monetary Equivalents Assessment

One method for comparing multi-attribute alternatives is to subjectively assign monetary values to the non-monetary attributes. For example, the decision maker may determine that each additional year of expected lifetime is worth $500, medium safety is $4,000 better than low safety, and high safety is $6,000 better than low safety. Arbitrarily using Fleet as the base case with total equivalent cost of $10,000, Figure 2 shows costs and equivalent costs, in thousands of dollars, in rows 9:11. The negative entries for Lifetime and Safety correspond to positive benefits relative to the Fleet car's base case values.

Based on this method, the Egret is chosen. Sensitivity analysis, not shown here, would involve seeing how the choice depends on subjective equivalents different from the $500 per year lifetime and the $4,000 and $6,000 safety assessments.

Hammond et al. [3] describe another method involving even swaps that could be used to select the best alternative.

Figure 2 Monetary Equivalents for Non-Dominated Alternatives

	A	B	C	D	E	F	G
1		Non-Dominated Alternatives					
2	Attribute	Alta	Bulldog	Cruiser	Delta	Egret	Fleet
3	Cost	$20	$18	$16	$14	$12	$10
4	Lifetime, years	10	10	8	8	6	6
5	Safety rating	High	Medium	High	Medium	Medium	Low
6							
7		Non-Dominated Alternatives					
8	Attribute	Alta	Bulldog	Cruiser	Delta	Egret	Fleet
9	Cost	$20	$18	$16	$14	$12	$10
10	Lifetime, $	-$2	-$2	-$1	-$1	$0	$0
11	Safety, $	-$6	-$4	-$6	-$4	-$4	$0
12							
13	Equiv. Cost	$12	$12	$9	$9	$8	$10

Additive Utility Function

The additive multi-attribute utility function U includes individual utility functions U_i for each attribute x_i, usually scaled from 0 to 1, and weights w_i that reflect the decision maker's tradeoffs among the attributes.

$$U(x_1,x_2,x_3) = w_1 \cdot U_1(x_1) + w_2 \cdot U_2(x_2) + w_3 \cdot U_3(x_3), \text{ where } w_1 + w_2 + w_3 = 1 \tag{1}$$

Weights may be specified directly, as ratios, or using a swing weight procedure. Individual utility functions are assessed using the range of attribute values for the alternatives being considered.

The individual utility values for Cost and Lifetime shown in Figure 3 are based on proportional scores, corresponding to linear utility functions. For example, each thousand dollar difference in cost is associated with a 0.1 difference in utility. The utility values for Safety are subjective judgments. For example, the decision maker thinks that a change in Safety from Low to Medium achieves only two-thirds of the satisfaction associated with a change from Low to High.

Figure 3 Individual Utilities

	A	B	C	D	E	F	G
1		Non-Dominated Alternatives					
2	Attribute	Alta	Bulldog	Cruiser	Delta	Egret	Fleet
3	Cost	$20	$18	$16	$14	$12	$10
4	Lifetime	10	10	8	8	6	6
5	Safety	High	Medium	High	Medium	Medium	Low
6							
7	Assess individual utility for each attribute.						
8	Cost	U($20,000)=0, U($10,000)=1, linear					
9	Lifetime	U(6 years)=0, U(10 years)=1, linear					
10	Safety	U(Low)=0, U(Medium)=2/3, U(High)=1					
11							
12		Non-Dominated Alternatives					
13	Attribute	Alta	Bulldog	Cruiser	Delta	Egret	Fleet
14	Cost	0.000	0.200	0.400	0.600	0.800	1.000
15	Lifetime	1.000	1.000	0.500	0.500	0.000	0.000
16	Safety	1.000	0.667	1.000	0.667	0.667	0.000

Compared to the assessments for individual utility, the assessments for tradeoffs are usually much more difficult to make. The following sections focus on assessments of tradeoff weights and sensitivity analysis.

Weight Ratio Assessment

One method for measuring trade-offs among the conflicting objectives is to assess weight ratios. For example, the decision maker may judge that cost is five times as important as lifetime, which may be interpreted to mean that the change in overall satisfaction corresponding to a change in cost from $20,000 to $10,000 is five times the change in overall satisfaction corresponding to a change in lifetime from 6 years to 10 years. Similarly, the decision maker may judge that a $10,000 decrease in cost is one and a half times as satisfying as a change from a low to a high safety rating. The assessments are shown in cells J4:J5 in Figure 4.

Figure 4 Weight Ratio Assessment and Choice

	A	B	C	D	E	F	G	H	I	J
1		Non-Dominated Alternatives							Assess weight ratios.	
2	Attribute	Alta	Bulldog	Cruiser	Delta	Egret	Fleet			
3	Cost	0.000	0.200	0.400	0.600	0.800	1.000		Weight Ratio	Input
4	Lifetime	1.000	1.000	0.500	0.500	0.000	0.000		Cost/Lifetime	5.0
5	Safety	1.000	0.667	1.000	0.667	0.667	0.000		Cost/Safety	1.5
6										
7	Overall	0.464	0.452	0.625	0.613	0.667	0.536		Weights	
8									Cost	0.536
9	Max Value	0.667							Lifetime	0.107
10	Location	5							Safety	0.357
11	Choice	Egret								
12										
13	Choice	Egret								

With three attributes, the two assessed weight ratios determine two equations and the requirement that the weights sum to one determines a third equation. Using algebra, a solution for the three unknown weights is shown in cells J8:J10 in Figure 5..

The formula for overall utility in cell B7, with a relative reference to the attribute utilities in B3:B5 and an absolute reference to the weights in J8:J10, is copied to cells C7:G7.

The MAX worksheet function determines the maximum overall utility in B7:G7, the MATCH function determines the location of that maximum in B7:G7, and the INDEX function returns the alternative name located in B2:G2. The zero argument in the MATCH function is needed to specify that an exact match is required; the zero argument in the INDEX function is used as a placeholder and could be omitted in this application without affecting the results. Cell B13 combines these functions into a single formula.

Figure 5 Formulas for Weight Ratio Assessment and Choice

	A	B	H	I	J
1		Non-Dominated Alternatives		Assess weight ratios.	
2	Attribute	Alta			
3	Cost	0		Weight Ratio	Input
4	Lifetime	1		Cost/Lifetime	5
5	Safety	1		Cost/Safety	1.5
6					
7	Overall	=SUMPRODUCT(B3:B5,$J$8:$J$10)		Weights	
8				Cost	=1/(1/J4+1/J5+1)
9	Max Value	=MAX(B7:G7)		Lifetime	=J8/J4
10	Location	=MATCH(B9,B7:G7,0)		Safety	=J8/J5
11	Choice	=INDEX(B2:G2,0,B10)			
12					
13	Choice	=INDEX(B2:G2,0,MATCH(MAX(B7:G7),B7:G7,0))			

After deleting cells A9:B12, the single formula is in cell B9. The arrangement shown in Figure 6 is used for the remaining analyses.

Figure 6 Weight Ratio Choice for Sensitivity Analysis

	A	B	C	D	E	F	G
1		Non-Dominated Alternatives					
2	Attribute	Alta	Bulldog	Cruiser	Delta	Egret	Fleet
3	Cost	0.000	0.200	0.400	0.600	0.800	1.000
4	Lifetime	1.000	1.000	0.500	0.500	0.000	0.000
5	Safety	1.000	0.667	1.000	0.667	0.667	0.000
6							
7	Overall	0.464	0.452	0.625	0.613	0.667	0.536
8							
9	Choice	Egret					

Weight Ratio Sensitivity Analysis

The decision maker specified tradeoffs using weight ratios, so it is appropriate to see whether the choice is sensitive to changes in those assessed values. To construct a two-way data table for sensitivity analysis of the weight ratios as shown in Figures 7 and 8, enter a set of values in a row, N4:R4, and another set of values in a column, M5:M13. In the top left cell of the data table, M4, enter a formula for determining the data table's output values, =B9. (To improve the appearance of the table, cell M4 is formatted with a custom three-semicolon format so that the formula result is not displayed.) Select M4:R13. Choose Data | Table. In the Data Table dialog box, specify J4 as the Row Input Cell and J5 as the Column Input Cell. Click OK.

Figure 7 Coarse Two-Factor Sensitivity Analysis of Weight Ratios

	L	M	N	O	P	Q	R
1	Two-Factor Sensitivity Analysis						
2							
3			Cost/Lifetime Weight Ratio				
4			3.0	4.0	5.0	6.0	7.0
5	Cost/Safety	1.00	Cruiser	Cruiser	Cruiser	Cruiser	Cruiser
6	Weight	1.25	Cruiser	Egret	Egret	Egret	Egret
7	Ratio	1.50	Egret	Egret	Egret	Egret	Egret
8		1.75	Egret	Egret	Egret	Egret	Egret
9		2.00	Egret	Egret	Egret	Egret	Egret
10		2.25	Egret	Egret	Egret	Egret	Egret
11		2.50	Egret	Egret	Egret	Egret	Egret
12		2.75	Egret	Egret	Egret	Egret	Egret
13		3.00	Egret	Egret	Egret	Egret	Egret

Cell P7, corresponding to the original assessments, has a border. The data table is dynamic, so the macro view may be refined near the base-case assessments by specifying different input values.

Figure 8 Fine Two-Factor Sensitivity Analysis of Weight Ratios

	L	M	N	O	P	Q	R
1	Two-Factor Sensitivity Analysis						
2							
3			Cost/Lifetime Weight Ratio				
4			4.0	4.5	5.0	5.5	6.0
5	Cost/Safety	1.00	Cruiser	Cruiser	Cruiser	Cruiser	Cruiser
6	Weight	1.10	Cruiser	Cruiser	Cruiser	Egret	Egret
7	Ratio	1.20	Cruiser	Egret	Egret	Egret	Egret
8		1.30	Egret	Egret	Egret	Egret	Egret
9		1.40	Egret	Egret	Egret	Egret	Egret
10		1.50	Egret	Egret	Egret	Egret	Egret
11		1.60	Egret	Egret	Egret	Egret	Egret
12		1.70	Egret	Egret	Egret	Egret	Egret
13		1.80	Egret	Egret	Egret	Egret	Egret

Figure 8 shows that the Cost/Safety weight ratio must be less than 1.2 to affect the choice. If the decision maker regards 1.2 as "far away" from 1.5, then the Egret choice is appropriate. Otherwise, the decision maker should think more carefully about the original assessments before making a choice based on this analysis. The assessment of the Cost/Lifetime weight ratio is not as critical, because any value between 4 and 6 yields the same choice.

Swing Weight Assessment

Compared to weight ratio assessment, the swing weight method requires assessments that are similar to directly assigning an overall utility to an alternative. However, the hypothetical alternatives requiring assessment in this method are constructed so that it should be easier for the decision maker to assign overall utilities to them instead of to the actual alternatives.

The swing weight method involves four steps as shown in Figure 9.

1) Develop the hypothetical alternatives. The number of hypothetical alternatives equals the number of attributes plus one. The benchmark alternative in column J is worst for all attributes. Each other hypothetical alternative, shown in columns K, L, and M, has one attribute at best and all others at worst.

2) Rank the hypothetical alternatives, as shown in row 7. This is an intermediate step that facilitates assigning overall utilities.

3) Assign overall utility scores reflecting overall satisfaction for the hypothetical alternatives. The benchmark worst case has score zero, and the first-ranked alternative has score 100. Then assign level-of-satisfaction scores to the intermediate alternatives, as shown in cells L9 and M9.

4) Sum the scores, as shown in cell N9. In the additive utility function, the weight for each attribute equals the score divided by sum of the scores. (The algebra solution, not shown here, is based on the special zero and one individual utility values of the hypothetical alternatives.) Formulas are shown in Figure 10.

Figure 9 Hypothetical Alternatives and Weights for Swing Weight Assessment

	I	J	K	L	M	N
1		Hypothetical Alternatives				
2	Attribute	Worst	Best Cost	Best Lifetime	Best Safety	
3	Cost	$20	$10	$20	$20	
4	Lifetime	6	6	10	6	
5	Safety	Low	Low	Low	High	
6						
7	Rank	4	1	3	2	
8						Total
9	Overall Score	0	100	20	70	190
10						
11	Weight	0.000	0.526	0.105	0.368	
12						
13	Decision Maker's Inputs Underlined					

Figure 10 Formulas for Swing Weight Assessment

	I	J	K	L	M	N
1		Hypothetical Alternatives				
2	Attribute	Worst	Best Cost	Best Lifetime	Best Safety	
3	Cost	20	10	20	20	
4	Lifetime	6	6	10	6	
5	Safety	Low	Low	Low	High	
6						
7	Rank	4	1	3	2	
8						Total
9	Overall Score	0	100	20	70	=SUM(J9:M9)
10						
11	Weight	=J9/$N$9	=K9/$N$9	=L9/$N$9	=M9/$N$9	
12						
13	Decision Maker's Inputs Underlined					

The individual utility values are in a column, and the weights are in a row. The SUMPRODUCT function requires that the two arrays for its arguments have the same orientation, so the TRANSPOSE function converts the weights into a column format, as shown in Figure 11. The function in B7 must be array-entered; after typing the function, hold down Control and Shift while you press Enter.

Figure 11 Formulas for Swing Weight Choice

	A	B
1		Non-Dominated Alternatives
2	Attribute	Alta
3	Cost	0
4	Lifetime	1
5	Safety	1
6		
7	Overall	=SUMPRODUCT(B3:B5,TRANSPOSE($K$11:$M$11))
8		
9	Choice	=INDEX(B2:G2,0,MATCH(MAX(B7:G7),B7:G7,0))

Figure 12 Swing Weight Choice

	A	B	C	D	E	F	G
1		Non-Dominated Alternatives					
2	Attribute	Alta	Bulldog	Cruiser	Delta	Egret	Fleet
3	Cost	0.000	0.200	0.400	0.600	0.800	1.000
4	Lifetime	1.000	1.000	0.500	0.500	0.000	0.000
5	Safety	1.000	0.667	1.000	0.667	0.667	0.000
6							
7	Overall	0.474	0.456	0.632	0.614	0.667	0.526
8							
9	Choice	Egret					

Swing Weight Sensitivity Analysis

The decision maker specified tradeoffs using overall scores for the hypothetical alternatives, so it is appropriate to see whether the choice is sensitive to changes in those assessed values. Figure 13 shows the sensitivity for the Best-Lifetime score that was specified as 20 relative to the worst-case benchmark and the highest-ranked Best-Cost hypothetical alternative. The Best-Lifetime alternative is still ranked 3 as long as its score is between 0 and 70.

To improve the appearance of the sensitivity analysis tables in Figure 13, the output formula cells, R13 and T13, have a three-semicolon custom format.

Figure 13 Sensitivity Analysis of Swing Weight Best-Lifetime Score

	P	Q	R	S	T	U
1	Single-Factor Sensitivity Analysis					
2						
3	Best Lifetime Overall Score					
4	Base case Score is 20					
5	Rank 3 as long as Score is between 0 and 70					
6						
7		Output Formula in cell R13: =B9				
8		Data Table Column Input Cell: M9				
9						
10					Detail	
11		Best Lifetime			Best Lifetime	
12		Overall Score	Choice		Overall Score	Choice
13						
14		0	Egret		30	Egret
15		5	Egret		31	Egret
16		10	Egret		32	Egret
17		15	Egret		33	Egret
18	Base Case	20	Egret		34	Cruiser
19		25	Egret		35	Cruiser
20		30	Egret			
21		35	Cruiser			
22		40	Cruiser			
23		45	Cruiser			
24		50	Cruiser			
25		55	Cruiser			
26		60	Cruiser			
27		65	Cruiser			
28		70	Cruiser			

The results in the left table Figure 13, cells Q13:R28, indicate that the Best-Lifetime score must be greater than 30 to affect the choice. A refined data table in cells T13:U19 shows that the score must be greater than 33 before the choice changes from Egret to Cruiser. If the decision maker regards 33 as "far away" from 20, then the Egret choice is appropriate.

Figure 14 shows a similar sensitivity analysis for the Best-Safety score. The assessed score of 70 must be greater than 89 to affect the choice.

Figure 14 Sensitivity Analysis of Swing Weight Best-Safety Score

	W	X	Y	Z	AA	AB
1	Single-Factor Sensitivity Analysis					
2						
3	Best Safety Overall Score					
4	Base case Score is 70					
5	Rank 2 as long as Score is between 20 and 100					
6						
7		Output Formula in cell Y13 and cell AB13: =B9				
8		Data Table Column Input Cell: N9				
9						
10					Detail	
11		Best Safety			Best Safety	
12		Overall Score	Choice		Overall Score	Choice
13						
14		20	Fleet		85	Egret
15		25	Fleet		86	Egret
16		30	Fleet		87	Egret
17		35	Egret		88	Egret
18		40	Egret		89	Egret
19		45	Egret		90	Cruiser
20		50	Egret			
21		55	Egret			
22		60	Egret			
23		65	Egret			
24	Base Case	70	Egret			
25		75	Egret			
26		80	Egret			
27		85	Egret			
28		90	Cruiser			
29		95	Cruiser			
30		100	Cruiser			

To construct a two-way data table for sensitivity analysis of the swing weight assessments as shown in Figure 15, enter a set of values in a row, R4:V4, and another set of values in a column, Q5:Q13. In the top left cell of the data table, Q4, enter a formula for determining the data table's output values, =B9. (To improve the appearance of the table, cell Q4 is formatted with a custom three-semicolon format so that the formula result is not displayed.) Select Q4:V13. Choose Data | Table. In the Data Table dialog box, specify L9 as the Row Input Cell and M9 as the Column Input Cell. Click OK.

Figure 15 Sensitivity Analysis of Both Swing Weight Scores

	P	Q	R	S	T	U	V
1	Two-Way Sensitivity Analysis						
2							
3			Best Lifetime Overall Score				
4			10	15	20	25	30
5	Best	50	Egret	Egret	Egret	Egret	Egret
6	Safety	55	Egret	Egret	Egret	Egret	Egret
7	Overall	60	Egret	Egret	Egret	Egret	Egret
8	Score	65	Egret	Egret	Egret	Egret	Egret
9		70	Egret	Egret	Egret	Egret	Egret
10		75	Egret	Egret	Egret	Egret	Cruiser
11		80	Egret	Egret	Egret	Egret	Cruiser
12		85	Egret	Egret	Egret	Cruiser	Cruiser
13		90	Egret	Egret	Cruiser	Cruiser	Cruiser

The table shows that the choice changes from Egret to Cruiser if the combination of assessments is changed from 20 & 70 to 30 & 75. This table could be refined to examine the exact threshold values.

Direct Weight Assessment and Sensitivity Analysis

In some situations the decision maker may be able to assign tradeoff weights directly. Figure 16 shows results using the formulas shown in Figure 17.

Figure 16 Direct Weight Assessment

	A	B	C	D	E	F	G	H	I	J
1		Non-Dominated Alternatives							Weights	
2	Attribute	Alta	Bulldog	Cruiser	Delta	Egret	Fleet		Cost	0.500
3	Cost	0.000	0.200	0.400	0.600	0.800	1.000		Lifetime	0.100
4	Lifetime	1.000	1.000	0.500	0.500	0.000	0.000		Safety	0.400
5	Safety	1.000	0.667	1.000	0.667	0.667	0.000			
6										
7	Overall	0.500	0.467	0.650	0.617	0.667	0.500			
8										
9	Choice	Egret								

The formula in cell B9 includes an IF function to verify that each weight is between 0 and 1, inclusive, and that the sum of the weights equals one. If not, the formula returns empty text. This formula must be array-entered; after typing the function, hold down Control and Shift while you press Enter.

Figure 17 Formulas for Direct Weight Assessment

	A	B	H	I	J
1		Non-Dominated Alternatives		Weights	
2	Attribute	Alta		Cost	0.5
3	Cost	0		Lifetime	0.1
4	Lifetime	1		Safety	=1-J3-J2
5	Safety	1			
6					
7	Overall	=SUMPRODUCT(B3:B5,$J$2:$J$4)			
8					
9	Choice	=IF(AND(SUM(J2:J4)<=1,J2:J4>=0),INDEX(B2:G2,0,MATCH(MAX(B7:G7),B7:G7,0)),"")			

Figure 18 shows a two-way table for sensitivity analysis of the weights. Cell R5 corresponds to the approximate base case assessments in the weight ratio and swing weight methods.

Figure 18 Sensitivity Analysis of Direct Weight Assessment

	L	M	N	O	P	Q	R	S	T	U	V
1	Two-Factor Sensitivity Analysis										
2											
3			Cost Weight								
4			0.1	0.2	0.3	0.4	0.5	0.6	0.7	0.8	0.9
5	Lifetime	0.1	Alta	Cruiser	Cruiser	Cruiser	Egret	Egret	Fleet	Fleet	Fleet
6	Weight	0.2	Alta	Alta	Cruiser	Cruiser	Cruiser	Egret	Fleet	Fleet	
7		0.3	Alta	Alta	Alta	Cruiser	Delta	Fleet	Fleet		
8		0.4	Alta	Alta	Alta	Bulldog	Bulldog	Fleet			
9		0.5	Alta	Alta	Alta	Bulldog	Bulldog				
10		0.6	Alta	Alta	Bulldog	Bulldog					
11		0.7	Alta	Bulldog	Bulldog						
12		0.8	Alta	Bulldog							
13		0.9	Bulldog								

Figure 19 is a more detailed view. The choice formula in cell B9 is modified by placing the INDEX function inside the LEFT function so that only the first letter of the alternative's name is returned.

Figure 19 Detailed Sensitivity Analysis of Direct Weight Assessment

	L	M	N	O	P	Q	R	S	T	U	V	W	X	Y	Z	AA	AB	AC	AD	AE	AF	AG	AH
1	Two-Factor Sensitivity Analysis																						
2																							
3			Cost Weight																				
4			0.00	0.05	0.10	0.15	0.20	0.25	0.30	0.35	0.40	0.45	0.50	0.55	0.60	0.65	0.70	0.75	0.80	0.85	0.90	0.95	1.00
5	Lifetime	0.00	A	C	C	C	C	C	C	C	C	C	E	E	E	E	E	E	F	F	F	F	F
6	Weight	0.05	A	A	C	C	C	C	C	C	C	C	E	E	E	E	E	F	F	F	F	F	
7		0.10	A	A	A	C	C	C	C	C	C	C	E	E	E	E	F	F	F	F	F		
8		0.15	A	A	A	A	C	C	C	C	C	C	E	E	E	E	F	F	F	F			
9		0.20	A	A	A	A	A	A	C	C	C	C	C	E	E	F	F	F	F				
10		0.25	A	A	A	A	A	A	A	C	C	C	D	D	F	F	F	F					
11		0.30	A	A	A	A	A	A	A	A	C	D	D	D	F	F	F						
12		0.35	A	A	A	A	A	A	A	A	A	D	D	D	F	F							
13		0.40	A	A	A	A	A	A	A	A	B	B	B	D	F								
14		0.45	A	A	A	A	A	A	A	B	B	B	B	B									
15		0.50	A	A	A	A	A	A	A	B	B	B	B										
16		0.55	A	A	A	A	A	A	B	B	B	B											
17		0.60	A	A	A	A	A	A	B	B	B												
18		0.65	A	A	A	A	A	B	B	B													
19		0.70	A	A	A	A	B	B	B														
20		0.75	A	A	A	A	B	B															
21		0.80	A	A	A	B	B																
22		0.85	A	A	B	B																	
23		0.90	A	A	B																		
24		0.95	A	B																			
25		1.00	A																				

The results in Figure 19 show that all alternatives in this data set are candidates depending on the tradeoffs specified by the decision maker. In general, moving left to right, if more weight is given to cost, a less expensive alternative is chosen.

Summary

This paper considered three methods for assessing tradeoffs in the additive utility function. For each method sensitivity analysis is useful for gaining insight into which tradeoff assumptions are critical. Kirkwood [2] includes Excel VBA methods for sensitivity analysis of individual utility functions in addition to weights.

REFERENCES

[1] Clemen, R.T. *Making Hard Decisions: An Introduction to Decision Analysis, 2nd Edition*. Duxbury Press, 1996.

[2] Kirkwood, C.W. *Strategic Decision Making: Multiobjective Decision Analysis with Spreadsheets*. Duxbury Press, 1997.

[3] Hammond, J.S., Keeney, R.L., and Raiffa, H. *Smart Choices: A Practical Guide to Making Better Decisions*. Harvard Business School Press, 1999.

Note

To copy Excel displays for the figures in this paper, choose File | Page Setup | Sheet | Gridlines and File | Page Setup | Sheet | Row And Column Headings. Select the cell range, hold down the Shift key, and in Excel's main menu choose Edit | Copy Picture | As Shown When Printed. In Word, position the pointer in an empty paragraph and choose Edit | Paste.